```
begin
    import Pkg
    # activate the shared project environment
    Pkg.activate(Base.current_project())
    using Omega, Distributions, UnicodePlots, OmegaExamples
    using Images, Plots
end
```

```
  Activating project at `~/Documents/GitHub/Omega.jl/OmegaExamples`
```

An important worry about Bayesian models of learning is that the Hypothesis space must either be too simple (e.g. a single coin weight!), specified in a rather ad-hoc way, or both. There is a tension here: human representations of the world are enormously complex and so the space of possible representations must be correspondingly big, and yet we would like to understand the representational resources in simple and uniform terms. How can we construct very large (possibly infinite) hypothesis spaces and priors over them? One possibility is to build the hypotheses themselves via stochastic recursion. That is, we build hypotheses by a combination of primitives and combination operations, randomly deciding which to use.

For instance, imagine that we want a model that generates strings, but we want the strings to be valid arithmetic expressions. Since we know that arithmetic has as primitives numbers and combines them with operations, we can define a simple generator:

```
random_const (generic function with 1 method)
```

```
random_const(i, ω::Ω) = string(((@uid, i) ~ UniformDraw(0:9))(ω))
```

```
random_combination (generic function with 1 method)
```

```
function random_combination(f, g, ω, i)
    op = ((@uid, i) ~ UniformDraw(['+', '-', '*', '/', '^']))(ω)
    return string('(', f, op, g, ')')
end
```

```
random_arithmetic_expression (generic function with 2 methods)
```

```
function random_arithmetic_expression(ω, i = 0)
    if ((@uid, i) ~ Bernoulli())(ω)
        e1 = random_arithmetic_expression(ω, (i..., 1))
        e2 = random_arithmetic_expression(ω, (i..., 2))
        return random_combination(e1, e2, ω, i)
    else
        return (i ~ random_const)(ω)
    end
end
```

```
"2"
```

```
randsample(random_arithmetic_expression)
```

Notice that `random_arithmetic_expression` can generate an infinite set of different strings, but that more complex strings are less likely. That is, the process we use to build strings also (implicitly)

defines a prior over strings that penalizes complexity. To see this more let's sample 100 strings:

```
(5-(9^(((((((9/(6+9))^(((3^9)-1)-(4+4)))*(7*2))^1)/((((((7+(((((8^3)/6)*1)^((7^7)+(9*(7*2))))
```

```
viz(randsample(random_arithmetic_expression, 100))
```

If we now interpret our strings as *hypotheses*, we have compactly defined an infinite hypothesis space and its prior.

Inferring an Arithmetic Function

Consider the following program, which induces an arithmetic function from examples. The basic form is the same as the above example but to evaluate the expression we use `eval(Meta.parse(x))` where `x` is the expression in string form.

```
evaluates_to_3 (generic function with 1 method)
```

```
function evaluates_to_3(ω::Ω)
    result = try
        eval(Meta.parse(random_arithmetic_expression(ω))) == 3
    catch
        false
    end
    return result
end
```

```
("(8+(((6-(((2-0)/(9*1))-9))^3)-(9^9)))", false)
```

```
randsample((random_arithmetic_expression, evaluates_to_3))
```

```
function_eval =
  Conditional(random_arithmetic_expression (generic function with 2 methods), evaluates_to_3
```

```
function_eval =
    random_arithmetic_expression |ᶜ evaluates_to_3
```


```
viz(randsample(function_eval, 50))
```

This model can learn any function consisting of the integers 0 to 9 and the operations add, subtract, multiply, divide, and power. The condition, in this case, asks for an arithmetic expression such that it evaluates to 3. There are many extensionally equivalent ways to satisfy the condition, for instance, the expressions `3`, `1 + 2`, but because the more complex expressions require more choices to generate, they are chosen less often.

Notice that the model puts the most probability on a function that always returns `3`. This is the simplest hypothesis consistent with the data.

Example: Rational Rules

How can we account for the productivity of human concepts (the fact that every child learns a remarkable number of different, complex concepts)? The “classical” theory of concepts formation accounted for this productivity by hypothesizing that concepts are represented compositionally, by logical combination of the features of objects (see for example Bruner, Goodnow, and Austin, 1951). That is, concepts could be thought of as rules for classifying objects (in or out of the concept) and concept learning was a process of deducing the correct rule.

While this theory was appealing for many reasons, it failed to account for a variety of categorization experiments. Here are the training examples, and one transfer example, from the classic experiment of Medin and Schaffer (1978). The bar graph above the stimuli shows the portion of human participants who said that bug was a “fep” in the test phase (the data comes from a replication by Nosofsky, Gluck, Palmeri, McKinley (1994); the bug stimuli are courtesy of Pat Shafto):



Notice three effects: there is a gradient of generalization (rather than all-or-nothing classification), some of the Feps are better (or more typical) than others (this is called "typicality"), and the transfer item is a "better" Fep than any of the Fep exemplars (this is called "prototype enhancement"). Effects like these were difficult to capture with classical rule-based models of category learning, which led to deterministic behavior. As a result of such difficulties, psychological models of category learning turned to more uncertain, prototype and exemplar based theories of concept representation. These models were able to predict behavioral data very well, but lacked compositional conceptual structure.

Is it possible to get graded effects from rule-based concepts? Perhaps these effects are driven by uncertainty in *learning* rather than uncertainty in the representations themselves? To explore these questions Goodman, Tenenbaum, Feldman, and Griffiths (2008) introduced the Rational Rules model, which learns deterministic rules by probabilistic inference. This model has an infinite hypothesis space of rules (represented in propositional logic), which are generated compositionally. Here is a slightly simplified version of the model, applied to the above experiment:

```
num_features = 4
```

```
1 num_features = 4
```

```
make_obj (generic function with 1 method)
```

```
1 make_obj(l) = NamedTuple(Dict(zip([:trait1, :trait2, :trait3, :trait4, :fep], l)))
```

```
feps =
```
```
[(trait3 = 0, trait1 = 0, fep = 1, trait4 = 1, trait2 = 0), (trait3 = 0, trait1 = 0, fep = 1, tr
```
```
feps = map(make_obj,
    [   [0, 0, 0, 1, 1],
        [0, 1, 0, 1, 1],
        [0, 1, 0, 0, 1],
        [0, 0, 1, 0, 1],
        [1, 0, 0, 0, 1]
    ])
```

```
non_feps =
```
```
[(trait3 = 1, trait1 = 0, fep = 0, trait4 = 1, trait2 = 0), (trait3 = 0, trait1 = 1, fep = 0, tr
```
```
non_feps = map(make_obj,
    [
        [0, 0, 1, 1, 0],
        [1, 0, 0, 1, 0],
        [1, 1, 1, 0, 0],
        [1, 1, 1, 1, 0]])
```

```
others =
```
```
[(trait3 = 1, trait1 = 0, trait4 = 0, trait2 = 1), (trait3 = 1, trait1 = 0, trait4 = 1, trait2 =
```
```
others = map(make_obj,
    [   [0, 1, 1, 0],
        [0, 1, 1, 1],
        [0, 0, 0, 0],
        [1, 1, 0, 1],
        [1, 0, 1, 0],
        [1, 1, 0, 0],
        [1, 0, 1, 1]
    ])
```

```
data =
```
```
[(trait3 = 0, trait1 = 0, fep = 1, trait4 = 1, trait2 = 0), (trait3 = 0, trait1 = 0, fep = 1, tr
```
```
data = vcat(feps, non_feps)
```

```
all_objects =
```
```
[(trait3 = 1, trait1 = 0, trait4 = 0, trait2 = 1), (trait3 = 1, trait1 = 0, trait4 = 1, trait2 =
```
```
all_objects = vcat(others, feps, non_feps)
```

```
[0.77, 0.78, 0.83, 0.64, 0.61, 0.39, 0.41, 0.21, 0.15, 0.56, 0.41, 0.82, 0.4, 0.32, 0.53, 0.2
```
```
begin
# here are the human results from Nosofsky et al, for comparison:
    human_feps = [.77, .78, .83, .64, .61]
    human_non_feps = [.39, .41, .21, .15]
    human_other = [.56, .41, .82, .40, .32, .53, .20]
    human_data = vcat(human_feps, human_non_feps, human_other)
end
```

```
τ = 0.2
```
```
τ = 0.2
```

```
noise_param = 0.22313016014842982
```
```
noise_param = exp(-1.5)
```

```
sample_pred (generic function with 1 method)
```

```
# a generative process for disjunctive normal form propositional equations:
function sample_pred(i, ω)
    trait = ((@uid, i...) ~ UniformDraw([:trait1, :trait2, :trait3, :trait4]))(ω)
    value = ((@uid, i...) ~ Bernoulli())(ω)
  return x -> (x[trait] == value)
end
```

```
sample_conj (generic function with 2 methods)
```

```
function sample_conj(ω, τ, i = 0)
    if ((@uid, :conj, i...) ~ Bernoulli(τ))(ω)
        c = sample_conj(ω, τ, i + 1)
        p = sample_pred(i, ω)
        return x -> (c(x) & p(x))
    else
        return sample_pred(i, ω)
    end
end
```

```
x =  (trait1 = true, trait2 = true, trait3 = true, trait4 = true, fep = 1)
```

```
x = (trait1 = true, trait2 = true, trait3 = true, trait4 = true, fep = 1)
```

```
get_formula (generic function with 2 methods)
```

```
function get_formula(ω, τ, i = 0)
    if ((@uid, :formula, i...) ~ Bernoulli(τ))(ω)
        c = sample_conj(ω, τ, i + 1)
        f = get_formula(ω, τ, i + 1)
        return x -> (c(x) | f(x))
    else
        return sample_conj(ω, τ, @uid)
    end
end
```

```
obs_fn (generic function with 1 method)
```

```
obs_fn(x, ω) =
    (@~ Bernoulli(ifelse.(get_formula(ω, τ)(x), 1 - noise_param, noise_param)))(ω)
```

```
evidence = ᵛ#7
```

```
evidence = Variable(ω -> all(map(x -> (obs_fn(x, ω) == (x.fep == 1)), data)))
```

```
rule_posterior =  Conditional(#11 (generic function with 1 method), ᵛ#7)
```

```
rule_posterior =  (ω -> get_formula(ω, τ, @uid)) |ᶜ evidence
```

```
samples =
  Vector{Bool}[
     1:  [false, false, true, false, true, false, false, false, false,  more ,false]
     2:  [false, false, true, false, true, false, false, false, false,  more ,false]
     3:  [false, false, true, false, true, false, false, false, false,  more ,false]
     4:  [false, false, true, false, false, false, false, true, true,  more ,false]
     5:  [false, false, true, false, false, false, false, true, true,  more ,false]
     6:  [false, false, true, false, false, false, false, true, true,  more ,false]
     7:  [false, false, true, false, true, false, false, false, false,  more ,false]
     8:  [false, false, true, false, false, false, false, true, true,  more ,false]
     9:  [true, true, true, false, false, false, false, true, true,  more ,false]
     10:  [false, false, true, false, false, false, false, true, true,  more ,false]
     11:  [true, true, true, false, true, false, true, true, true,  more ,true]
     12:  [false, false, true, false, false, false, false, true, true,  more ,false]
     13:  [false, true, true, true, false, false, true, true, true,  more ,true]
     14:  [false, false, true, false, true, false, false, false, false,  more ,false]
     15:  [false, false, true, true, true, true, true, true, true,  more ,true]
     16:  [false, false, true, false, false, false, false, true, true,  more ,false]
     17:  [false, false, true, true, true, true, true, true, true,  more ,true]
     18:  [false, true, true, true, true, false, true, true, true,  more ,true]
     19:  [false, false, true, false, false, false, false, true, true,  more ,false]
     20:  [false, false, true, false, false, false, false, true, true,  more ,false]
    more
     91:  [false, false, true, false, false, false, false, true, true,  more ,false]
     92:  [false, false, true, false, true, false, false, false, false,  more ,false]
     93:  [false, false, true, false, true, false, false, false, false,  more ,false]
     94:  [false, false, true, false, false, false, false, true, true,  more ,false]
     95:  [false, false, true, false, false, false, false, true, true,  more ,false]
     96:  [true, true, true, false, true, true, true, false, false,  more ,true]
     97:  [false, false, true, false, false, false, false, true, true,  more ,false]
     98:  [false, false, true, false, false, false, false, true, true,  more ,false]
     99:  [false, false, true, true, false, true, false, true, true,  more ,false]
     100:  [false, false, true, false, false, false, false, true, true,  more ,false]
]
```

```
1 samples = randsample(ω -> map(rule_posterior(ω), all_objects), 100)
```




```
1 scatterplot(first(mean(samples, dims=1)), human_data, marker = :xcross)
```

In addition to achieving a good overall correlation with the data, this model captures the three qualitative effects described above: graded generalization, typicality, and prototype enhancement. Make sure you see how to read each of these effects from the above plot! Goodman, et al, have used to this model to capture a variety of other classic categorization effects (Goodman et al., 2008), as well. Thus probabilistic induction of (deterministic) rules can capture many of the graded effects previously taken as evidence against rule-based models.

Grammar-based induction

What is the general principle in the two above examples? We can think of it as the following recipe: we build hypotheses by stochastically choosing between primitives and combination operations, this specifies an infinite "language of thought"; each expression in this language in turn specifies the likelihood of observations. Formally, the stochastic combination process specifies a probabilistic grammar; which yields terms compositionally interpreted into a likelihood over data. A small grammar can generate an infinite array of potential hypotheses; because grammars are themselves generative processes, a prior is provided for free from this formulation.

This style of compositional concept induction model, can be naturally extended to complex hypothesis spaces, each defined by a grammar. For instance to model theory acquisition, learning natural numbers concepts, and many others. See:

- Compositionality in rational analysis: Grammar-based induction for concept learning. N. D. Goodman, J. B. Tenenbaum, T. L. Griffiths, and J. Feldman (2008). In M. Oaksford and N. Chater (Eds.). The probabilistic mind: Prospects for Bayesian cognitive science.
- A Bayesian Model of the Acquisition of Compositional Semantics. S. T. Piantadosi, N. D. Goodman, B. A. Ellis, and J. B. Tenenbaum (2008). Proceedings of the Thirtieth Annual Conference of the Cognitive Science Society.
- Piantadosi, S. T., & Jacobs, R. A. (2016). Four Problems Solved by the Probabilistic Language of Thought. Current Directions in Psychological Science, 25(1).